Problem:
Score : 300 points

Problem Statement

You drew lottery N times. In the i-th draw, you got an item of the kind represented by a string S_i.

How many kinds of items did you get?

Constraints

1 \leq N \leq 2\times 10^5

S_i consists of lowercase English letters and has a length between 1 and 10 (inclusive).

Input

Input is given from Standard Input in the following format:

N
S_1
:
S_N

Output

Print the number of kinds of items you got.

Sample Input 1

3
apple
orange
apple

Sample Output 1

2

You got two kinds of items: apple and orange.

Sample Input 2

5
grape
grape
grape
grape
grape

Sample Output 2

1

Sample Input 3

4
aaaa
a
aaa
aa

Sample Output 3

4

Solution in Fortran (GNU Fortran 9.2.1) (judge verdict: Accepted)

program main
    use,intrinsic :: iso_fortran_env
    implicit none
    integer(int32):: n,i,j,ans
    character(10),allocatable:: s(:)


    read*, n
    allocate(s(n))
    do i=1,n
        read*, s(i)
    end do
    call character_sort(s,1,n)
    ans=1
    do i=2,n
        if (s(i) /= s(i-1)) ans=ans+1
    end do
    print*, ans

contains
    recursive subroutine character_sort(ar, fst, lst)
        character(10),intent(inout):: ar(:)
        integer(int32),intent(in):: fst,lst
        integer(int32):: mdl

        if (lst-fst < 2) then
            if (ar(fst) > ar(lst)) call swap(ar(fst),ar(lst))
            return
        end if

        mdl = (fst+lst)/2
        call character_sort(ar, fst, mdl)
        call character_sort(ar, mdl+1, lst)
        call merge_(ar, fst, mdl, lst)
    end subroutine

    subroutine merge_(ar, fst, mdl, lst)
        character(10),intent(inout):: ar(:)
        integer(int32),intent(in):: fst, mdl, lst
        character(10),allocatable:: tmp(:)
        integer(int32):: li, ri, ti

        allocate(tmp(lst-fst+1))

        li=fst
        ri=mdl+1   
        ti=1

        do while (li <= mdl .and. ri <= lst)
            if (ar(li) <= ar(ri)) then
                tmp(ti) = ar(li)
                li=li+1
            else
                tmp(ti) = ar(ri)
                ri=ri+1
            end if
            ti=ti+1
        end do

        if (li <= mdl) then
            tmp(ti:) = ar(li:mdl)
        else
            tmp(ti:) = ar(ri:lst)
        end if

        ar(fst:lst) = tmp(:)
        deallocate(tmp)
    end subroutine

    subroutine swap(x,y)
        character(10),intent(inout):: x,y
        character(10):: tmp
        tmp = x
        x = y
        y = tmp
    end subroutine
end program